
The Primitive Euclidean Algorithm and Intermediate Expression Swell.
Michael Monagan, Spring 2021.

restart;
a := 3*x^3+x^2+x+5;

a 3 x3 x2 x 5
b := 5*x^2-3*x+1;

b 5 x2 3 x 1

Pseudo division in ℤ[x]:  prem( a, b, x ) = rem( lc(b)^(deg(a) - deg(b) + 1), a, b, x ) = rem( 5^2 a, b, x )
prem(a,b,x);

52 x 111
rem(5^2*a,b,x);

52 x 111
pr := prem(a,b,x,'m','pq');

pr 52 x 111
m, pq;

25, 15 x 14
expand( m*a=b*pq+pr );

75 x3 25 x2 25 x 125 = 75 x3 25 x2 25 x 125

An example of pseudo-division in ℤ[y][x]
a := 5*x^3+y*x^2+3;

a 5 x3 y x2 3
b := 3*y^2*x^2 + 5*x + 7*y; 

b 3 y2 x2 5 x 7 y

In this example the multiplier m = lc a 2 = 32 y4

pr := prem(a,b,x,'m');
pr 120 x y3 6 y4 125 x 175 y

m;
9 y4

rem(m*a,b,x);
120 y3 125  x 6 y4 175 y

The Primitive Euclidean Algorithm
g := 2*x-3;

g 2 x 3
a := expand( g*randpoly(x,degree=4,dense) );

a 14 x5 65 x4 176 x3 23 x2 456 x 261
b := expand( g*randpoly(x,degree=4,dense) );

b 112 x5 168 x4 124 x3 380 x2 437 x 219
content(a,x);

1
content(b,x);

1
r[0] := primpart(a,x);

r[1] := primpart(b,x);
k := 1;
while r[k] <> 0 do
    pr := prem(r[k-1],r[k],x);
    r[k+1] := primpart(pr,x);
    k := k+1;
od;
g := r[k-1];
g := gcd(content(a,x),content(b,x))*g; # attach gcd of contents

r0 14 x5 65 x4 176 x3 23 x2 456 x 261

r1 112 x5 168 x4 124 x3 380 x2 437 x 219

k 1

pr 4928 x4 17976 x3 7896 x2 57190 x 32298

r2 352 x4 1284 x3 564 x2 4085 x 2307

k 2

pr 146318080 x3 160375552 x2 200787904 x 168203328

r3 2286220 x3 2505868 x2 3137311 x 2628177

k 3

pr 5568818946161152 x2 12794178883681536 x 6661425696659712

r4 179778504202 x2 413035217061 x 215051191137

k 4
pr 14164158620635468468339428800 x 21246237930953202702509143200

r5 2 x 3

k 5
pr 0
r6 0

k 6
g 2 x 3
g 2 x 3

We need to multiply g by 1 to make it unit normal in ℤ[x].
Notice that the degree of the pseudo-remainder pr is decreasing until we get pr = gcd BUT the size of the 
integer coefficients are growing.  Let us spy on the size of the integer coefficients for a larger example.
a := expand( g*randpoly(x,degree=20,dense) );
b := expand( g*randpoly(x,degree=20,dense) );

a 8 x21 154 x20 229 x19 154 x18 350 x17 406 x16 328 x15 274 x14

247 x13 99 x12 205 x11 16 x10 59 x9 204 x8 226 x7 196 x6 81 x5

409 x4 288 x3 130 x2 78 x 216

b 74 x21 157 x20 243 x19 173 x18 74 x17 109 x16 340 x15 89 x14

152 x13 167 x12 34 x11 277 x10 263 x9 131 x8 53 x7 221 x6 282 x5

425 x4 137 x3 224 x2 92 x 93
r[0] := primpart(a,x):
r[1] := primpart(b,x):
k := 1:
while r[k] <> 0 do
    printf("k=%2d  deg=%2d  size=%d\n",
            k,degree(r[k],x),length(maxnorm(r[k])) );
    pr := prem(r[k-1],r[k],x);
    r[k+1] := primpart(pr,x);
    k := k+1;
od:
g := r[k-1];
g := gcd(content(a,x),content(b,x))*g; # attach gcd of contents

k= 1  deg=21  size=3
k= 2  deg=20  size=5
k= 3  deg=19  size=9
k= 4  deg=18  size=13
k= 5  deg=17  size=17
k= 6  deg=16  size=21
k= 7  deg=15  size=25
k= 8  deg=14  size=29
k= 9  deg=13  size=33
k=10  deg=12  size=39
k=11  deg=11  size=43
k=12  deg=10  size=48
k=13  deg= 9  size=53
k=14  deg= 8  size=57
k=15  deg= 7  size=61
k=16  deg= 6  size=65
k=17  deg= 5  size=71
k=18  deg= 4  size=75
k=19  deg= 3  size=79
k=20  deg= 2  size=84
k=21  deg= 1  size=1

g 2 x 3
g 2 x 3

Notice that the size of the coefficients increases by just over 4 digits each iteration.  This is a linear 
growth in the size of the coefficients.  The degree is dropping but coefficients grow until we get to the 
gcd.
I'm going to redo the example with two variables instead of one so for ℤ[x,y] and we are going to watch 
what happens to the degree of the pseudo-remainders in y.

g := 2*x-3*y-5;
g 2 x 3 y 5

a := expand( g*randpoly([x,y],degree=10,dense) );
a 102 x11 307 x10 y 229 x9 y2 59 x8 y3 54 x7 y4 139 x6 y5 237 x5 y6

286 x4 y7 251 x3 y8 253 x2 y9 388 x y10 141 y11 445 x10 668 x9 y

24 x8 y2 26 x7 y3 72 x6 y4 141 x5 y5 113 x4 y6 279 x3 y7 219 x2 y8

376 x y9 352 y10 365 x9 110 x8 y 362 x7 y2 669 x6 y3 265 x5 y4 15 x4 y5

255 x3 y6 293 x2 y7 22 x y8 354 y9 329 x8 75 x7 y 448 x6 y2 481 x5 y3

143 x4 y4 432 x3 y5 380 x2 y6 194 x y7 481 y8 39 x7 281 x6 y 161 x5 y2

184 x4 y3 327 x3 y4 433 x2 y5 436 x y6 651 y7 259 x6 699 x5 y 201 x4 y2

515 x3 y3 497 x2 y4 468 x y5 386 y6 276 x5 492 x4 y 171 x3 y2 16 x2 y3

229 x y4 195 y5 314 x4 164 x3 y 310 x2 y2 53 x y3 71 y4 240 x3

459 x2 y 392 x y2 232 y3 142 x2 73 x y 250 y2 331 x 361 y 60
b := expand( g*randpoly([x,y],degree=9,dense) );

b 50 x10 117 x9 y 168 x8 y2 2 x7 y3 147 x6 y4 2 x5 y5 201 x4 y6 67 x3 y7

202 x2 y8 159 x y9 108 y10 225 x9 246 x8 y 286 x7 y2 687 x6 y3 98 x5 y4

718 x4 y5 220 x3 y6 95 x2 y7 206 x y8 453 y9 236 x8 257 x7 y 352 x6 y2

98 x5 y3 108 x4 y4 138 x3 y5 108 x2 y6 19 x y7 389 y8 171 x7 6 x6 y

40 x5 y2 391 x4 y3 208 x3 y4 353 x2 y5 347 x y6 43 y7 332 x6 168 x5 y

424 x4 y2 641 x3 y3 159 x2 y4 526 x y5 174 y6 64 x5 142 x4 y 216 x3 y2

12 x2 y3 363 x y4 15 y5 388 x4 15 x3 y 333 x2 y2 150 x y3 430 y4

365 x3 86 x2 y 236 x y2 27 y3 96 x2 340 x y 596 y2 454 x 526 y
485

r[0] := primpart(a,x):
r[1] := primpart(b,x):
k := 1:
while r[k] <> 0 do
    printf("k=%2d  degx=%2d  degy=%3d  size=%d\n",
            k,degree(r[k],x),degree(r[k],y),length(maxnorm(r[k])) )
;
    pr := prem(r[k-1],r[k],x);
    r[k+1] := primpart(pr,x);
    k := k+1;
od:
g := r[k-1];
g := gcd(content(a,x),content(b,x))*g; # attach gcd of con

k= 1  degx=10  degy= 10  size=3
k= 2  degx= 9  degy= 11  size=7
k= 3  degx= 8  degy= 14  size=12
k= 4  degx= 7  degy= 19  size=16
k= 5  degx= 6  degy= 26  size=21
k= 6  degx= 5  degy= 35  size=26
k= 7  degx= 4  degy= 46  size=31
k= 8  degx= 3  degy= 59  size=36
k= 9  degx= 2  degy= 74  size=41
k=10  degx= 1  degy=  1  size=1

g 2 x 3 y 5
g 2 x 3 y 5

There is now a two dimensional linear growth.  The degree in y is growing and the size of the integer 
coefficients is also growing.  This means the intermediate pseudo-remainders, before we get to the gcd, 
can be much larger than the input polynomials.  Let us make the example larger and time Maple executing 
the primitive Euclidean algorithm and then time Maple doing the gcd using whatever algorithm it uses.

a := expand( g*randpoly([x,y],degree=40,dense) ):

b := expand( g*randpoly([x,y],degree=39,dense) ):
st := time():
r[0] := primpart(a,x):
r[1] := primpart(b,x):
k := 1:
while r[k] <> 0 do
    printf("k=%2d  degx=%2d  degy=%3d  size=%d\n",
            k,degree(r[k],x),degree(r[k],y),length(maxnorm(r[k])) )
;
    pr := prem(r[k-1],r[k],x);
    r[k+1] := primpart(pr,x);
    k := k+1;
od:
g := r[k-1];
g := gcd(content(a,x),content(b,x))*g; # attach gcd of contents
time=time()-st;

k= 1  degx=40  degy= 40  size=3
k= 2  degx=39  degy= 41  size=7
k= 3  degx=38  degy= 44  size=12
k= 4  degx=37  degy= 49  size=16
k= 5  degx=36  degy= 56  size=21
k= 6  degx=35  degy= 65  size=26
k= 7  degx=34  degy= 76  size=32
k= 8  degx=33  degy= 89  size=37
k= 9  degx=32  degy=104  size=42
k=10  degx=31  degy=121  size=48
k=11  degx=30  degy=140  size=54
k=12  degx=29  degy=161  size=59
k=13  degx=28  degy=184  size=65
k=14  degx=27  degy=209  size=71
k=15  degx=26  degy=236  size=76
k=16  degx=25  degy=265  size=82
k=17  degx=24  degy=296  size=88
k=18  degx=23  degy=329  size=94
k=19  degx=22  degy=364  size=101
k=20  degx=21  degy=401  size=107
k=21  degx=20  degy=440  size=113
k=22  degx=19  degy=481  size=119
k=23  degx=18  degy=524  size=125
k=24  degx=17  degy=569  size=132
k=25  degx=16  degy=616  size=137
k=26  degx=15  degy=665  size=144
k=27  degx=14  degy=716  size=150
k=28  degx=13  degy=769  size=156
k=29  degx=12  degy=824  size=163
k=30  degx=11  degy=881  size=169
k=31  degx=10  degy=940  size=176
k=32  degx= 9  degy=1001  size=182
k=33  degx= 8  degy=1064  size=188
k=34  degx= 7  degy=1129  size=195
k=35  degx= 6  degy=1196  size=202
k=36  degx= 5  degy=1265  size=208
k=37  degx= 4  degy=1336  size=214
k=38  degx= 3  degy=1409  size=220
k=39  degx= 2  degy=1484  size=226
k=40  degx= 1  degy=  1  size=1

g 2 x 3 y 5
g 2 x 3 y 5

time = 346.809

st := time(): gcd(a,b); time()-st;
2 x 3 y 5

0.024

Maple is NOT using the primitive Euclidean algorithm!  If we added a third variable z to this example, 
Maple might never finish.  Basically, the complexity of the primitive Euclidean algorithm is exponential in
the number of variables.
In the 1970s polynomial gcd computation was a major area of research because it was impossible to 
compute the gcd of two polynomials in many variables for even small degrees.  This phenomenon where 
the intermediate expressions are much larger than the inputs and output(s) has been termed intermediate 
expression swell.  It also occurs in Gaussian elimination when applied to a matrix of integers or 
polynomials.


